﻿
CREATE PROCEDURE [dbo].[LedgerOperationsVsTransTypesUpdate]
(--список типов продводок, которые экспортируются в 1С для заданного типа операции
	@LedgerOperationId int,
	@LedgerTransTypeId int,
	@Note varchar(250),
	@MD datetime,
	@MW varchar(50),
	@ToDelete bit,
	@Original_Id int,
	@Original_MD datetime
)
AS
DECLARE @strMsg nvarchar(500), @intRet int, @intErrCode int
DECLARE @orgOperationId int, @orgTransTypeId int, @orgToDelete bit, @orgParentToDelete bit

SET NOCOUNT ON

IF (@LedgerOperationId Is Null) OR (@LedgerTransTypeId Is Null) OR (@ToDelete Is Null)
BEGIN
	SELECT @intErrCode = 1, @strMsg = 'Неправильные параметры в вызове процедуры ' + Object_Name(@@Procid) + ' !'
	RAISERROR (@strMsg, 11, @intErrCode)
	RETURN @intErrCode
END

SELECT @orgOperationId = LedgerOperationId, @orgTransTypeId = LedgerTransTypeId, @orgToDelete = ToDelete
FROM [CO].LedgerOperationsVsTransTypes WHERE Id = @Original_Id AND MD = @Original_MD
IF (@@rowcount <> 1)
BEGIN
	SELECT @intErrCode = 2, 
	@strMsg = 'Данные изменены после последнего запроса!'
	RAISERROR (@strMsg, 11, 1)
	RETURN @intErrCode
END

IF  (@orgOperationId <> @LedgerOperationId) OR (@orgTransTypeId <> @LedgerTransTypeId)
IF EXISTS (SELECT TOP 1 ID FROM [CO].LedgerOperationsVsTransTypes 
WHERE LedgerOperationId = @LedgerOperationId 
AND LedgerTransTypeId = @LedgerTransTypeId 
AND ToDelete = 0 )
BEGIN
	SELECT @intErrCode = 2, @strMsg = 'Дублируется тип проводки для операции!'
	RAISERROR (@strMsg, 11, @intErrCode)
	RETURN @intErrCode
END

SELECT @orgParentToDelete = ToDelete
FROM CO.LedgerOperations WHERE Id = @LedgerOperationId
IF (@@rowcount <> 1)
BEGIN
	SELECT @intErrCode = 3, 
	@strMsg = 'Отсутствует родительская запись с кодом ' + cast(@LedgerOperationId as varchar(20)) + ' в справочнике операций!'
	RAISERROR (@strMsg, 11, 1)
	RETURN @intErrCode
END

IF @orgParentToDelete = 1 
BEGIN
	SELECT @intErrCode = 4, 
	@strMsg = 'Родительская запись с кодом ' + cast(@LedgerOperationId as varchar(20)) + ' в справочнике операций отмечена для удаления!'
	RAISERROR (@strMsg, 11, 1)
	RETURN @intErrCode
END

SELECT @orgParentToDelete = ToDelete
FROM CO.LedgerTransTypes WHERE Id = @LedgerTransTypeId
IF (@@rowcount <> 1)
BEGIN
	SELECT @intErrCode = 5, 
	@strMsg = 'Отсутствует родительская запись с кодом ' + cast(@LedgerTransTypeId as varchar(20)) + ' в справочнике типов проводок!'
	RAISERROR (@strMsg, 11, 1)
	RETURN @intErrCode
END

IF @orgParentToDelete = 1 
BEGIN
	SELECT @intErrCode = 6, 
	@strMsg = 'Родительская запись с кодом ' + cast(@LedgerTransTypeId as varchar(20)) + ' в справочнике типов проводок отмечена для удаления!'
	RAISERROR (@strMsg, 11, 1)
	RETURN @intErrCode
END

IF @orgToDelete = 1 --Запись была отмечена для удаления
BEGIN
	IF @ToDelete = 1 
	BEGIN
		SELECT @intErrCode = 7, 
		@strMsg = 'Нельзя редактировать запись, отмеченную для удаления!'
		RAISERROR (@strMsg, 11, 1)
		RETURN @intErrCode
	END
END ELSE BEGIN
	IF @ToDelete = 1 
	BEGIN
		SELECT @intErrCode = 8, 
		@strMsg = 'Для удаления данных должна использоваться другая процедура!'
		RAISERROR (@strMsg, 11, 1)
		RETURN @intErrCode
	END
END

EXEC [dbo].[ProcessStatistic] @Note OUTPUT, @MD OUTPUT, @MW OUTPUT
	SET NOCOUNT OFF;
	
UPDATE [CO].LedgerOperationsVsTransTypes SET 
LedgerOperationId = @LedgerOperationId, LedgerTransTypeId = @LedgerTransTypeId,
[Note] = @Note, [MD] = @MD, [MW] = @MW, ToDelete = @ToDelete 
WHERE Id = @Original_Id AND MD = @Original_MD;


SELECT LedgerOperationId, LedgerTransTypeId, Note, MD, MW, ToDelete 
FROM CO.LedgerOperationsVsTransTypes WHERE (Id = @Original_Id)

RETURN 0
